Solution to Exercise 2.6.2

Clearly $|G| = 2^4 \cdot 4! = 384$ and it is easy to find generators for G. We use GAP:

```
gap> a := [ [0,1,0,0], [-1,0,0,0], [0,0,1,0], [0,0,0,1] ];;
gap> b := [ [0,1,0,0], [0,0,1,0], [0,0,0,1], [1,0,0,0] ];;
gap> G := Group(a,b);; Size(G);
384
gap> G1 := Group( Filtered( Elements(G), x -> Determinant(x) = 1) );;
gap> G2 := Group( Filtered( Elements(G), x -> Permanent(x) = 1) );;
```

Instead of computing the character tables of the matrix groups G_1 = `G1` and G_2 = `G2` we first transform them into permutation groups (of degree 16) using their actions on $C := \{(a_1, a_2, a_3, a_4) \mid a_i \in \{1, -1\}\}$. This will speed up the computations from minutes to seconds.

```
gap> C := Cartesian( [1,-1], [1,-1],[ 1,-1], [1,-1] );;
gap> G1 := Image( ActionHomomorphism(G1,C) );;
gap> G2 := Image( ActionHomomorphism(G2,C) );;
gap> ct1 := CharacterTable(G1);;  ct2 := CharacterTable(G2);;
gap> ct1 := CharacterTableWithSortedClasses(ct1);;
gap> ct2 := CharacterTableWithSortedClasses(ct2);;
gap> per := TransformingPermutations( Irr(ct1), Irr(ct2) );
rec( columns := (4,9,6,8,5,10,7)(11,13), rows := (6,9,7,8),
  group := Group([ (6,7), (4,5)(11,12), (3,4)(10,11) ]) )
gap> Display(ct1);
CT1


      2  6  6  5  4  3  1  5  5  4  4  1  3  3
      3  1  1  .  .  .  1  .  .  .  .  1  .  .

        1a 2a 2b 2c 2d 3a 4a 4b 4c 4d 6a 8a 8b

X.1      1  1  1  1  1  1  1  1  1  1  1  1  1
X.2      1  1  1  1 -1  1  1  1 -1 -1  1 -1 -1
X.3      2  2  2  2  . -1  2  2  .  . -1  .  .
X.4      3  3  3 -1 -1  . -1 -1 -1 -1  .  1  1
X.5      3  3  3 -1  1  . -1 -1  1  1  . -1 -1
X.6      3  3 -1 -1 -1  .  3 -1  1  1  . -1  1
X.7      3  3 -1 -1 -1  . -1  3  1  1  .  1 -1
X.8      3  3 -1 -1  1  . -1  3 -1 -1  . -1  1
X.9      3  3 -1 -1  1  .  3 -1 -1 -1  .  1 -1
X.10     4 -4  .  .  .  1  .  . -2  2 -1  .  .
X.11     4 -4  .  .  .  1  .  .  2 -2 -1  .  .
X.12     6  6 -2  2  .  . -2 -2  .  .  .  .  .
X.13     8 -8  .  .  . -1  .  .  .  .  1  .  .

gap> OrdersClassRepresentatives(ct2);
[ 1, 2, 2, 2, 2, 2, 2, 3, 4, 4, 4, 4, 6 ]
```

G_1 has elements of order 8 while the maximal order of an element of G_2 is 6, so it is clear that G_1 and G_2 don't form a Brauer pair. But the command `TransformingPermutations` has found permutations which aplied to the rows and columns of $\mathrm{Irr}(G_2)$ gives the matrix $\mathrm{Irr}(G_1)$. We check this:

```
gap> pc := per.columns;  pr := per.rows;
(4,9,6,8,5,10,7)(11,13)
(6,9,7,8)
gap> ForAll( [1..13], i -> Irr(ct1)[i] =
>                       List([1..13], j -> Irr(ct2)[i^pr][j^pc]) );
true
```

Finally we compute the normal subgroups of G_1 and G_2 and check that G_2 has three abelian normal subgroups of order 8 while G_1 has only one such normal subgroup, which is, in fact, $\ker$ `X.4`, as the displayed character table reveals.

```
gap> List( NormalSubgroups(G1), N -> [Size(N), IsAbelian(N)] );
[ [ 1, true ], [ 2, true ], [ 8, false ], [ 8, false ], [ 8, true ],
  [ 32, false ], [ 96, false ], [ 192, false ] ]
gap> List( NormalSubgroups(G2), N -> [Size(N), IsAbelian(N)] );
[ [ 1, true ], [ 2, true ], [ 8, true ], [ 8, true ], [ 8, true ],
  [ 32, false ], [ 96, false ], [ 192, false ] ]
```

Remark. Let, more generally, P_n be the group of all $n \times n$-permutation matrices, $D_n := \{\mathrm{diag}(a_1, \ldots, a_n) \mid a_i \in \{1, -1\}\}$ and $G_n := D_n P_n$. Furthermore $G_n^{(1)} := \{A \in G_n \mid \det(A) = 1\}$ and $G_n^{(2)} := \{A \in G_n \mid \mathrm{per}(A) = 1\}$. In [134] it is proved that for $n \equiv 0 \mod 4$ the groups $G_n^{(1)}$ and $G_n^{(2)}$ have the same character table, while $G_n^{(2)}$ which is also known as the "Weyl group of type D_n" is determined by its character table for $n \not\equiv 0 \mod 4$.